

Extreme Learning Machine in J

Pierre-Edouard Portier

2019

1 Regression

$\mathbf{x}^{(1)} \dots \mathbf{x}^{(P)}$ are vectors of \mathbb{R}^{n-1} with associated values $y^{(1)} \dots y^{(P)}$ of \mathbb{R} . We search a function $f(\mathbf{x}) : \mathbb{R}^{n-1} \rightarrow \mathbb{R}$ to model the observed relationship between \mathbf{x} and y . f can have a fixed parameterized form. For example:

$$f(\mathbf{x}) = a_0 + a_1x_1 + a_2x_2 + \dots + a_{n-1}x_{n-1}$$

If $P = n$, parameters $a_0 \dots a_{n-1}$ are found by solving a linear system.

$$\begin{cases} y^{(1)} &= a_0 + a_1x_1^{(1)} + a_2x_2^{(1)} + \dots + a_{n-1}x_{n-1}^{(1)} \\ \dots &= \dots \\ y^{(P)} &= a_0 + a_1x_1^{(P)} + a_2x_2^{(P)} + \dots + a_{n-1}x_{n-1}^{(P)} \end{cases}$$

This system can be written in matrix form.

$$\begin{pmatrix} 1 & x_1^{(1)} & \dots & x_{n-1}^{(1)} \\ 1 & x_1^{(2)} & \dots & x_{n-1}^{(2)} \\ \dots & \dots & \dots & \dots \\ 1 & x_1^{(P)} & \dots & x_{n-1}^{(P)} \end{pmatrix} \begin{pmatrix} a_0 \\ a_1 \\ \dots \\ a_{n-1} \end{pmatrix} = \begin{pmatrix} y^{(1)} \\ y^{(2)} \\ \dots \\ y^{(P)} \end{pmatrix}$$

Each line of the first term matrix is a vector $\mathbf{x}^{(i)T}$ with the addition of a constant coordinate that accounts for parameter a_0 . Thus, naming this matrix \mathbf{X}^T , the linear system can also be written:

$$\mathbf{X}^T \mathbf{a} = \mathbf{y}$$

Consider the special case when x is a number and f is a polynomial of degree $n - 1$:

$$f(x) = a_0 + a_1x + a_2x^2 + \dots + a_{n-1}x^{n-1}$$

With $P = n$ examples $(x^{(k)}, y^{(k)})$, the parameters are found by solving the following linear system:

$$\begin{pmatrix} 1 & x^{(1)} & (x^{(1)})^2 & \dots & (x^{(1)})^{n-1} \\ 1 & x^{(2)} & (x^{(2)})^2 & \dots & (x^{(2)})^{n-1} \\ \dots & \dots & \dots & \dots & \dots \\ 1 & x^{(P)} & (x^{(P)})^2 & \dots & (x^{(P)})^{n-1} \end{pmatrix} \begin{pmatrix} a_0 \\ a_1 \\ \dots \\ a_{n-1} \end{pmatrix} = \begin{pmatrix} y^{(1)} \\ y^{(2)} \\ \dots \\ y^{(P)} \end{pmatrix} \quad (1)$$

Incidentally, the first term is called the Vandermonde Matrix.

1.1 Experiment with a 1-dimensional synthetic dataset

We define a non linear function f from which we generate a dataset

```
2a <dataset 2a>≡ (9b)
    f=: 3 : ' (^y) * cos 2*pi * sin pi * y'
    <noise 2b>
    <gendat 2d>
```

In traditional mathematical form, this function is:

$$f(x) = e^x \times \cos(2\pi \sin(\pi x))$$

Function `noise` adds some random noise to the values of a vector. For example `0.5 noise v`, will add random values uniformly drawn from interval $[-0.5, 0.5]$ to the terms of vector v .

```
2b <noise 2b>≡ (2a)
    noise=: 4 : 'y + -&x *&(+:x) ? (#y) # 0'
```

`0.5 gendat 10` generates from f a dataset (X, Y) of 10 points with random noise in $[-0.5, 0.5]$ added to Y . It also stores in `minmaxX` the minimum and maximum values of X . It computes the pair `minmaxf`, where the first term is ten percent smaller than the minimum of f on interval $[0, 1]$, and the second term is ten percent bigger than the maximum of f on interval $[0, 1]$. `minmaxf` is later used to crop the plots so that extreme values are not visible.

```
2c <utils 2c>≡ (9b) 3d>
    pushup=: ] + 0.1 * |
    pushdown=: ] - 0.1 * |
```

```
2d <gendat 2d>≡ (2a)
    gendat=: 4 : 0
    X=: ? y $ 0
    Y=: x noise f X
    minmaxX=: (<./ , >./) X
    minmaxf=: (([: pushdown <./) , ([: pushup >./)) f steps 0 1 100
    <testdat 8a>
    )
```

`plotdat 0` plots the dataset.

```
2e <plotdat 2e>≡ (9b)
    plotdatnoshow=: 3 : 0
    <initplot 3a>
    pd X;Y
    <plotf 3b>
    )
    plotdat=: 3 : 0
    plotdatnoshow 0
    pd 'show'
    )
```

```


3a <initplot 3a>≡ (2e 8d)
    pd 'reset'
    pd 'color green'
    pd 'type marker'
    pd 'markersize 1'
    pd 'markers circle'

```

```

3b <plotf 3b>≡ (2e 8d)
    pd 'color red'
    pd 'type line'
    pd 'pensize 1'
    pd (;f) steps 0 1 100

```



```

0.5 gendat 10
_2.53128 2.99011
plotdat 0

```

polyreg 0 solves the linear system (1) and stores the coefficients of the polynomial in variable c.

```

3c <polyreg 3c>≡ (9b)
    polyreg=: 3 : 0
    c=: Y ([ %. ] ^/ i.@#0] X
    plotpoly 0
)

```

```


3d <utils 2c>+≡ (9b) <2c 5b>
    NB. select from y the elements with values between {x and {x
    sel=: ([ >: {.@[ ] *. ([ <: {:@[ ]

```

```

3e <plotpoly 3e>≡ (9b)
    plotpoly=: 3 : 0
    plotdatnoshow 0
    pd 'color blue'
    xs=: ([ #~ minmaxX"_ sel ]) /:~ X,steps 0 1 100
    pval=: c&p. xs
    crop=: minmaxf sel pval
    pd (crop # xs);(crop # pval)
    pd 'show'
)

```



polyreg 0

1.2 Generalization to a function space

Given a basis for a function space, we can try to express \mathbf{f} as a combination of basis functions.

$$f(\mathbf{x}) = a_1 f_1(\mathbf{x}) + a_2 f_2(\mathbf{x}) + \dots + a_n f_n(\mathbf{x})$$

Given a dataset of n pairs $(\mathbf{x}^{(k)}, \mathbf{y}^{(k)})$, the coefficients a_i are found by solving a linear system.

$$\begin{pmatrix} f_1(\mathbf{x}^{(1)}) & f_2(\mathbf{x}^{(1)}) & \dots & f_n(\mathbf{x}^{(1)}) \\ f_1(\mathbf{x}^{(2)}) & f_2(\mathbf{x}^{(2)}) & \dots & f_n(\mathbf{x}^{(2)}) \\ \dots & \dots & \dots & \dots \\ f_1(\mathbf{x}^{(n)}) & f_2(\mathbf{x}^{(n)}) & \dots & f_n(\mathbf{x}^{(n)}) \end{pmatrix} \begin{pmatrix} a_1 \\ a_2 \\ \dots \\ a_n \end{pmatrix} = \begin{pmatrix} y^{(1)} \\ y^{(2)} \\ \dots \\ y^{(n)} \end{pmatrix}$$

Let us denote this linear system by $\mathbf{Ax} = \mathbf{b}$.

1.3 Least squares

With more examples than the number of basis functions, the linear system $\mathbf{Ax} = \mathbf{b}$ (with $\mathbf{A} \in \mathbb{R}^{m \times n}$) doesn't necessarily have a solution. Thus, we want to find an approximate solution $\mathbf{Ax} \approx \mathbf{b}$ that minimizes the squares of the errors: $\|\mathbf{Ax} - \mathbf{b}\|_2^2$.

$$\begin{aligned} & \|\mathbf{Ax} - \mathbf{b}\|_2^2 \\ = & \{ \|\mathbf{x}\|_2 = \sqrt{\mathbf{x} \cdot \mathbf{x}} \} \\ & (\mathbf{Ax} - \mathbf{b}) \cdot (\mathbf{Ax} - \mathbf{b}) \\ = & \{ \text{euclidean scalar product} \} \\ & (\mathbf{Ax} - \mathbf{b})^T (\mathbf{Ax} - \mathbf{b}) \\ = & \{ \text{property of transposition} \} \\ & (\mathbf{x}^T \mathbf{A}^T - \mathbf{b}^T) (\mathbf{Ax} - \mathbf{b}) \\ = & \{ \text{multiplication} \} \\ & \mathbf{x}^T \mathbf{A}^T \mathbf{Ax} - \mathbf{x}^T \mathbf{A}^T \mathbf{b} - \mathbf{b}^T \mathbf{Ax} + \mathbf{b}^T \mathbf{b} \\ = & \{ \text{Since each element of the sum is a scalar, } \mathbf{b}^T \mathbf{Ax} = (\mathbf{b}^T \mathbf{Ax})^T = \mathbf{x}^T \mathbf{A}^T \mathbf{b} \} \\ & \mathbf{x}^T \mathbf{A}^T \mathbf{Ax} - 2\mathbf{x}^T \mathbf{A}^T \mathbf{b} + \mathbf{b}^T \mathbf{b} \end{aligned}$$

To this quadratic expression corresponds a convex surface. Its minimum is found by setting the derivative to zero.

$$\mathbf{0} = 2\mathbf{A}^T \mathbf{A} \mathbf{x} - 2\mathbf{A}^T \mathbf{b}$$

$$=$$

$$\mathbf{A}^T \mathbf{A} \mathbf{x} = \mathbf{A}^T \mathbf{b}$$

Thus, when $m > n$, we solve $\mathbf{A} \mathbf{x} \approx \mathbf{b}$ by solving $\mathbf{A}^T \mathbf{A} \mathbf{x} = \mathbf{A}^T \mathbf{b}$. $\mathbf{A}^T \mathbf{A}$ is called the Gram matrix. `gram y` computes the Gram matrix \mathbf{S} for a polynomial basis of degree $y-1$.

5a `<gram 5a>≡` (9b) <5c>

```
gram=: 3 : 0
  A=: X ^/ i.y
  S=: (mp~ |:) A
)
```


5b `<utils 2c>+≡` (9b) <3d 6b>

```
mp=: +/ . * NB. matrix product
```


`leastsq y` solves the overdetermined linear system by computing the Gram matrix for a polynomial basis of degree $y-1$.

5c `<gram 5a>+≡` (9b) <5a>

```
leastsq=: 3 : 0
  gram y
  c=: ((|:A) mp Y) %. S
  plotpoly 0
)
```



```
0.5 gendat 100
_2.53128 2.99011
leastsq 5
```



```
leastsq 8
```

1.4 Tikhonov regularization

With less examples than the number of basis functions (i.e. $m < n$, underdetermined system), $\mathbf{Ax} = \mathbf{b}$ doesn't have a unique solution. Even with $m \geq n$, the linear system can have approximate solutions more desirable than the optimal one. In particular, this is the case when several examples are very similar. For example, the solution to...

$$\begin{pmatrix} 1 & 1 \\ 1 & 1.00001 \end{pmatrix} \begin{pmatrix} x_1 \\ x_2 \end{pmatrix} = \begin{pmatrix} 1 \\ 0.99 \end{pmatrix}$$

... is $\mathbf{x}^T = (1001, -1000)$. However, the approximate solution $\mathbf{x}^T = (0.5, 0.5)$ is more suitable. Indeed, the optimal solution is not likely to adapt well to new inputs (e.g., input (1, 2) would be projected onto $-999...$).

Thus, when several solutions are feasible, we want to favor smaller norms $\|\mathbf{x}\|_2$ by solving a new minimization problem:

$$\min_{\mathbf{x}} \|\mathbf{Ax} - \mathbf{b}\|_2^2 + \alpha \|\mathbf{x}\|_2^2$$

with $0 < \alpha < 1$

The minimum of this expression is found by setting its derivative to zero.

$$\begin{aligned} \mathbf{0} &= 2\mathbf{A}^T \mathbf{Ax} - 2\mathbf{A}^T \mathbf{b} + 2\alpha \mathbf{x} \\ &= \\ &= \left(\mathbf{A}^T \mathbf{A} + \alpha \mathbf{I}_{n \times n} \right) \mathbf{x} = \mathbf{A}^T \mathbf{b} \end{aligned}$$

It comes down to adding a small positive value to the diagonal of the Gram matrix. This approach has been given several names: Tikhonov regularization, ridge regression...

1E.3 ridge 5 will solve the ridge regression for a polynomial basis of degree 5 and a regularization coefficient equal to 10^{-3} .

6a `<ridge 6a>≡` (9b)

```

ridge=: 4 : 0
  gram y
  c=: ((|:A) mp Y) %. x addDiag S
  plotpoly 0
)
```

6b `<utils 2c>+≡` (9b) <5b 8b>

```

diag=: (<0 1)&|: : ([[>:*i.)[:#]])
addDiag=: ([+diag@]) diag ] NB. add x to the diagonal of y
```

1.5 Extreme Learning Machine

The following parametrized form for f corresponds to a single hidden layer neural network.

$$f(\mathbf{x}) = a_1 g(\mathbf{w}_1 \cdot \mathbf{x} + b_1) + a_2 g(\mathbf{w}_2 \cdot \mathbf{x} + b_2) + \dots + a_M g(\mathbf{w}_M \cdot \mathbf{x} + b_M)$$

g is a non-linear activation function. We use the rectified linear unit (ReLU): $g(y) = \max(0, y)$.

If vectors $\mathbf{w}_1 \dots \mathbf{w}_M$ and scalars $b_1 \dots b_M$ are initialized randomly and never modified (i.e., if they are not parameters), we can solve a linear system $\mathbf{H}\mathbf{a} = \mathbf{y}$ of unknown \mathbf{a} .

$$\mathbf{H} : \begin{pmatrix} g(\mathbf{w}_1 \cdot \mathbf{x}_1 + b_1) & \dots & g(\mathbf{w}_M \cdot \mathbf{x}_1 + b_M) \\ \dots & \dots & \dots \\ g(\mathbf{w}_1 \cdot \mathbf{x}_N + b_1) & \dots & g(\mathbf{w}_M \cdot \mathbf{x}_N + b_M) \end{pmatrix}$$

$$\mathbf{a}^T : (a_1 \dots a_M)$$

$$\mathbf{y}^T : (y_1 \dots y_N)$$

This approach is named *Extreme Learning Machine*¹.

`initelm` 100 initializes randomly matrix \mathbf{H} with 100 neurons on the hidden layer (i.e., $M = 100$) and computes its Gram form \mathbf{S} .

7a `<elm 7a>`≡ (9b) 7b▷

```
initelm=: 3 : 0
  W=: _1 + 2 * ? (y,1) $ 0 NB. input weights
  B=: ? y $ 0 NB. bias
  H=: mkH ,. X
  0 [ S=: (mp~ |: ) H
)
mkH=: 3 : '0&>. B +"1 y mp"1/ W'
```

`elm` 1E-4 solves the extreme learning machine linear system with a Tikhonov regularization coefficient of 10^{-4} .


7b `<elm 7a>+≡` (9b) <7a

```
elm=: 3 : 0
  c=: ((|:H) mp Y) % . y addDiag S
  plotelm 0
)
```

7c `<plotelm 7c>`≡ (9b)

```
plotelm=: 3 : 0
  plotdatnoshow 0
  pd 'type line'
  pd 'color blue'
  xs=: (| #~ minmaxX" _ sel |) steps (<.<./X),(>.>./X),100
  pd xs;(mkH ,. xs) mp c
  pd 'show'
)
```

¹<https://scholar.google.fr/scholar?q=extreme+learning+machine>



```
initelm 100
0
elm 1E_3
```

1.6 Test dataset

A test set is used to assert the capacity of the model to generalize on unseen data. Its size is fixed to 10% of the size of the training set.


```
8a <testdat 8a>≡ (2d)
    XT=: ? (>. 0.1 * y) $ 0
    YT=: f XT
```

`test` computes the root mean square error (RMSE) on the test set.

```
8b <utils 2c>+≡ (9b) <6b
    mean=: +/ % #
    rmse=: [: %: [: mean ([: *: -)
```

```
8c <test 8c>≡ (9b)
    test=: 3 : 0
    YThat=: (mkH ,. XT) mp c
    plottest 0
    YT rmse YThat
)
```

```
8d <plottest 8d>≡ (9b)
    plottest=: 3 : 0
    <initplot 3a>
    pd XT;YT
    pd 'color magenta'
    pd XT;YThat
    <plotf 3b>
    pd 'show'
)
```

test 0
0.172019

9a `<require 9a>≡`
`require'trig'`
`require'plot'`
`require'numeric'`

(9b)

9b `<jelm.ijs 9b>≡`
`<require 9a>`
`<utils 2c>`
`<dataset 2a>`
`<plotdat 2e>`
`<plotpoly 3e>`
`<polyreg 3c>`
`<gram 5a>`
`<ridge 6a>`
`<plotelm 7c>`
`<elm 7a>`
`<plottest 8d>`
`<test 8c>`